

Matching Methods

John Doe

2026-01-31

Contents

Overview	5
1 Propensity Score Matching	7
1.1 Generate propensity score	7
1.2 Estimate the <i>ATE</i> and <i>ATT</i> without trimming	10
1.3 Trim the data	13
1.4 Check for balance again	14
1.5 Reestimate the <i>ATT</i> and <i>ATE</i>	14
1.6 Recheck for balance	15
2 Inverse Probability Weights	19
2.1 Manual Calculate the <i>ATE</i>	20
2.2 Calculate <i>ATE</i> and <i>ATT</i> using <i>teffects</i>	23
3 Doubly Robust Estimator	27
3.1 Calculate the <i>ATE</i> and <i>ATT</i> using Doubly Robust Estimator . .	28

bookdown::render_book()

Overview

Topics:

1. Propensity Score Matching
2. Inverse Probability Weights
3. Doubly Robust Estimator

Chapter 1

Propensity Score Matching

We will go through the process of manually estimating propensity scores before we use the *teffects* commands. After estimating the propensity scores, we will estimate the *ATE* and *ATT*

We will use the data from Dehejia and Wahba (2002) for treatment from Department of Labor Employment and Training Administration (ETA) job training. Instead of using the control group from the original RCT, we will use data from the Current Population Survey (CPS) for a comparison group.

1.1 Generate propensity score

First, we will import and drop the experimental control group, so we can append our comparison group from the CPS.

```
use https://github.com/scunning1975/mixtape/raw/master/nsw_mixtape.dta, clear  
drop if treat==0
```

Now merge in the CPS controls from footnote 2 of Table 2 (Dehejia and Wahba 2002)

```
append using https://github.com/scunning1975/mixtape/raw/master/cps_mixtape.dta
```

Now we will generate additional variables of interest: age, education, demographics, and unemployment status

```

gen agesq=age*age
gen agecube=age*age*age
gen edusq=educ*educ
gen u74 = 0 if re74!=.
replace u74 = 1 if re74==0
gen u75 = 0 if re75!=.
replace u75 = 1 if re75==0
gen interaction1 = educ*re74
gen re74sq=re74^2
gen re75sq=re75^2
gen interaction2 = u74*hispanic

```

Next, we will use logistic regression to estimate β and estimate the p -scores.

```

logit treat age agesq agecube educ edusq marr nodegree black hispanic re74 re75 u74 u75 in
predict pscore

```

```

Iteration 0:    log likelihood = -1011.0713
Iteration 1:    log likelihood = -612.55814
Iteration 2:    log likelihood = -429.05652
Iteration 3:    log likelihood = -406.25926
Iteration 4:    log likelihood = -404.18813
Iteration 5:    log likelihood = -404.15992
Iteration 6:    log likelihood = -404.15991

```

Logistic regression	Number of obs	=	16,177
	LR chi2(14)	=	1213.82
	Prob > chi2	=	0.0000
Log likelihood = -404.15991	Pseudo R2	=	0.6003

treat	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
age	2.425229	.3500651	6.93	0.000	1.739114	3.111344
agesq	-.0672395	.0111308	-6.04	0.000	-.0890555	-.0454234
agecube	.0005685	.0001113	5.11	0.000	.0003505	.0007866
educ	.9247846	.2500694	3.70	0.000	.4346576	1.414912
edusq	-.0572021	.0136202	-4.20	0.000	-.0838972	-.030507
marr	-1.556471	.2517687	-6.18	0.000	-2.049929	-1.063013
nodegree	.9270591	.3254621	2.85	0.004	.2891651	1.564953
black	3.850668	.2662868	14.46	0.000	3.328755	4.37258
hispanic	1.673885	.4099129	4.08	0.000	.8704704	2.4773
re74	-.0002203	.0001086	-2.03	0.043	-.0004332	-7.40e-06
re75	-.0001969	.0000378	-5.21	0.000	-.000271	-.0001228
u74	1.749522	.2897311	6.04	0.000	1.18166	2.317385

1.1. GENERATE PROPENSITY SCORE

9

```

      u75 |      .00944      .257531      0.04      0.971      -.4953114      .5141915
interaction1 | .0000222      9.08e-06      2.45      0.014      4.43e-06      .00004
      _cons | -35.22098      3.797922      -9.27      0.000      -42.66477      -27.77719
-----

```

Note: 3 failures and 0 successes completely determined.

(option pr assumed; Pr(treat))

Now, we will compare the propensity score between treatment and comparison groups with summary statistics and histograms.

```

sum pscore if treat==1, detail
sum pscore if treat==0, detail

```

Pr(treat)				


	Percentiles	Smallest		
1%	.0011757	.0010614		
5%	.0072641	.0011757		
10%	.0260147	.0018463	Obs	185
25%	.1322174	.0020981	Sum of Wgt.	185
50%	.4001992		Mean	.4253567
		Largest	Std. Dev.	.3076908
75%	.6706164	.9356451		
90%	.8866026	.93718	Variance	.0946736
95%	.9021386	.9374608	Skewness	.1916478
99%	.9374608	.9384554	Kurtosis	1.732134

Pr(treat)				

	Percentiles	Smallest		
1%	5.90e-07	1.18e-09		
5%	1.72e-06	4.07e-09		
10%	3.58e-06	4.24e-09	Obs	15,992
25%	.0000193	1.55e-08	Sum of Wgt.	15,992
50%	.0001187		Mean	.0066476
		Largest	Std. Dev.	.0416672
75%	.0009635	.8786677		
90%	.0066319	.8893389	Variance	.0017362
95%	.0163109	.9099022	Skewness	12.2493
99%	.1551548	.9239787	Kurtosis	187.8692

The comparison group has a mean propensity score of 0.01. The treatment group has a mean propensity score is 0.42

```
histogram pscore, by(treat) binrescale
```



It seems that there is very little overlap for the common support assumption. It would be problematic to estimate the *ATE* and *ATT* without trimming to get common support.

1.2 Estimate the *ATE* and *ATT* without trimming

We will estimate the *ATE* and *ATT* with the *teffects psmatch* command. We will use the 5 nearest neighbors with the option *nn(5)*. Note that Stata generates new variables defined by you. Here we are defining *pstub_cps*.

Estimate *ATT*

```
teffects psmatch (re78) (treat age agesq agecube educ edusq marr nodegree black ///
  hisp re74 re75 u74 u75 interaction1, logit), atet gen(pstub_cps) nn(5)
drop pstub_cps*
```

Treatment-effects estimation	Number of obs	=	16,177
Estimator : propensity-score matching	Matches: requested	=	5
Outcome model : matching	min	=	5

Treatment model: logit

max = 26

		AI Robust				
re78	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
ATET						
treat						
(1 vs 0)	1725.082	700.0586	2.46	0.014	352.9921	3097.172

$$\widehat{ATT} = \$1,725.082$$

We have a few more options for checking the balance after the PSM with nearest neighbor of 5. We can use the *tebalance* commands to check our balance. *tebalance density* will give us a distribution of propensity scores by our variable of choice between treatment and comparison. We can also use *tebalance box* that will give is a box-and-whisker chart for the propensity scores by our variable of choice between treatment and comparison.

```
tebalance summarize, baseline
```

Covariate balance summary

	Raw	Matched
Number of obs =	16,177	370
Treated obs =	185	185
Control obs =	15,992	185

	Means		Variances	
	Control	Treated	Control	Treated
age	33.22524	25.81622	121.9968	51.1943
agesq	1225.906	717.3946	615814.1	185978
agecube	49305.85	21554.66	2.04e+09	4.40e+08
educ	12.02751	10.34595	8.241754	4.042714
edusq	152.9023	111.0595	4511.316	1544.795
marr	.7117309	.1891892	.2051829	.1542303
nodegree	.2958354	.7081081	.2083299	.2078143
black	.0735368	.8432432	.0681334	.1329025
hisp	.072036	.0594595	.066851	.056228
re74	14016.8	2095.574	9.16e+07	2.39e+07
re75	13650.8	1532.055	8.59e+07	1.04e+07
u74	.1196223	.7081081	.1053194	.2078143
u75	.1093047	.6	.0973632	.2413043

```
interaction1 | 171147.6    22898.73    1.67e+10    3.29e+09
-----
```

After running `tebalance summarize`, we can see without trimming that our co-variables are not really balanced between treatment and comparison groups.

```
tebalance density age
```




Figure 1.1: Density Chart for Age

```
tebalance box age
drop pstub_cps*
```

Next, we will estimate the *ATE* with `teffects psmatch`.

```
teffects psmatch (re78) (treat age agesq agecube educ edusq marr nodegree black ///
  hisp re74 re75 u74 u75 interaction1, logit), ate gen(pstub_cps) nn(5)
drop pstub_cps*
```

We find that we have an error estimating the *ATE*, since we lack common support.




Figure 1.2: Box Chart for Age

1.3 Trim the data

We will use the trim the p -scores. All values below 10% and above 90% will be dropped.

```
drop if pscore <= 0.1
drop if pscore >= 0.9
```


We can see that the comparison group has a lot of observations with p -scores less than or equal to 0.1 (15,802), but we only drop 14 observations with p -scores greater than or equal 0.9.

```
(15,802 observations deleted)
```

```
(14 observations deleted)
```

1.4 Check for balance again

```
histogram pscore, by(treat) binrescale
graph export "pscore2.png", replace
```



We can now see we have better overlap between treatment and comparison after trimming the extreme values of the p -scores.

1.5 Reestimate the ATT and ATE

```
teffects psmatch (re78) (treat age agesq agecube educ edusq marr nodegree black ///
  hisp re74 re75 u74 u75 interaction1, logit), ate gen(pstub_cps) nn(5)

drop pstub_cps*
```

Treatment-effects estimation	Number of obs	=	361
Estimator : propensity-score matching	Matches: requested	=	5
Outcome model : matching	min	=	5
Treatment model: logit	max	=	8

	re78	Coef.	AI Robust Std. Err.	z	P> z	[95% Conf. Interval]	
ATE	treat						
	(1 vs 0)	1758.837	724.5845	2.43	0.015	338.6778	3178.997

$$\widehat{ATE} = \$1758.84$$

```
teffects psmatch (re78) (treat age agesq agecube educ edusq marr nodegree black ///
  hisp re74 re75 u74 u75 interaction1, logit), atet gen(pstub_cps) nn(5)
```

```
Treatment-effects estimation      Number of obs      =      361
Estimator      : propensity-score matching      Matches: requested =      5
Outcome model  : matching                      min =      5
Treatment model: logit                      max =      8
```

	re78	Coef.	AI Robust Std. Err.	z	P> z	[95% Conf. Interval]	
ATET	treat						
	(1 vs 0)	2519.446	866.7552	2.91	0.004	820.6375	4218.255

$$\widehat{ATE} = \$2519.446$$

1.6 Recheck for balance

```
tebalance summarize, baseline
```

Covariate balance summary

	Raw	Matched
Number of obs =	361	266
Treated obs =	133	133
Control obs =	228	133

		Means		Variances	
		Control	Treated	Control	Treated
age		25.93421	25.2406	64.98684	45.00228
agesq		737.2851	681.7519	237422	158499.2
agecube		22966.88	19825.29	6.03e+08	3.66e+08
educ		10.46491	10.2406	4.584667	3.850763
edusq		114.0789	108.6917	1830.681	1331.306
marr		.2192982	.1428571	.1719607	.1233766
nodegree		.6315789	.7518797	.233712	.1879699
black		.9342105	.9398496	.061732	.0569606
hisp		.0438596	.037594	.0421207	.0364548
re74		2552.555	1560.604	1.93e+07	1.56e+07
re75		1570.309	1002.193	7324536	3233917
u74		.5570175	.7669173	.247836	.1801094
u75		.4868421	.6390977	.2509274	.2323992
interaction1		27361.94	16425.15	2.44e+09	1.81e+09

Our balance is greatly improved after trimming the tail ends of the propensity scores.

Let's check the variable age with *tebalance density* and *tebalance box*.

```
tebalance density age
```

```
tebalance box age
```





Figure 1.3: Density Chart of Age After Trimming




Figure 1.4: Box Chart of Age After Trimming

Chapter 2

Inverse Probability Weights

We will reload our treatment group and CPS comparison group, and use the *teffects ipw* command to estimate the *ATE*.

We will repull our data, estimate our *p*-score, and check our balance

```
*Get Data
use https://github.com/scunning1975/mixtape/raw/master/nsw_mixtape.dta, clear
drop if treat==0
append using https://github.com/scunning1975/mixtape/raw/master/cps_mixtape.dta

*Calculate Variables
gen agesq=age*age
gen agecube=age*age*age
gen edusq=educ*educ
gen u74 = 0 if re74!=.
replace u74 = 1 if re74==0
gen u75 = 0 if re75!=.
replace u75 = 1 if re75==0
gen interaction1 = educ*re74
gen re74sq=re74^2
gen re75sq=re75^2
gen interaction2 = u74*hisp

*Estimate the coefficients
logit treat age agesq agecube educ edusq marr nodegree black hisp re74 re75 u74 u75 interaction1
predict pscore

* Checking mean propensity scores for treatment and control groups
sum pscore if treat==1, detail
sum pscore if treat==0, detail
```

```
* Now look at the propensity score distribution for treatment and control groups
histogram pscore, by(treat) binrescale
```




Figure 2.1: Histogram for IPW

2.1 Manual Calculate the *ATE*

We will manually calculate inverse probability weighting (IPW) with Propensity Scores. Please note that IPW is not matching, even though we use propensity scores.

IPW with Non-normalized Weights:

$$\widehat{ATE} = \frac{1}{n} * \sum_{n=1}^N \left(\frac{Y_i * D_i}{\hat{p}(x)_i} \right) - \frac{1}{n} * \sum_{n=1}^N \left(\frac{Y_i * (1 - D_i)}{1 - \hat{p}(x)_i} \right)$$

IPW with Normalized Weights:

$$\widehat{ATE} = \left[\sum_{n=1}^N \frac{Y_i * D_i}{\hat{p}(x)_i} \right] / \left[\sum_{n=1}^N \frac{D_i}{\hat{p}(x)_i} \right] - \left[\sum_{n=1}^N \frac{Y_i * (1 - D_i)}{(1 - \hat{p}(x)_i)} \right] / \left[\sum_{n=1}^N \frac{(1 - D_i)}{(1 - \hat{p}(x)_i)} \right]$$

2.1.1 Calculate $E[Y^1]$ and $E[Y^0]$

First, we need to calculate normalization weights

```
gen d1=treat/pscore
gen d0=(1-treat)/(1-pscore)
```

Sum the inverse pscore

```
egen s1=sum(d1)
egen s0=sum(d0)
gen total = _N
egen total_T = sum(treat)
egen total_C = sum(1-treat)
```

First, we will manually calculate $E[Y^1]$ with non-normalized weights using all the data. Calculate $E[Y^1] = \frac{1}{n} * \sum_{n=1}^N \frac{Y_i * D_i}{\hat{p}(x)_i}$

```
gen y1=treat*re78/pscore
egen y1_2 = sum(y1)
gen y1_3 = y1_2/total
```

Next, we will manually calculate $E[Y^0]$ with non-normalized weights using all the data. Calculate $E[Y^0] = \frac{1}{n} * \sum_{n=1}^N \frac{Y_i * (1-D_i)}{(1-\hat{p}(x)_i)}$

```
gen y0=(1-treat)*re78/(1-pscore)
egen y0_2 = sum(y0)
gen y0_3 = y0_2/total
```

Then calculate the *ATE* by taking the difference in $E[Y^1] - E[Y^0]$.

```
*Way 1
gen ht=y1-y0
*Way 2
gen ht2=y1_3-y0_3
*Same results
sum ht ht2
```

Variable	Obs	Mean	Std. Dev.	Min	Max
ht	16,177	-11876.79	127578.8	-99232.64	1.19e+07
ht2	16,177	-11876.79	0	-11876.79	-11876.79

Now we will calculate the *ATE* manually using normalized weights

```
*Way 1
replace y1=(treat*re78/pscore)/(s1/total)
replace y0=((1-treat)*re78/(1-pscore))/(s0/total)
*Way 2
replace y1_3=y1_2/s1
replace y0_3=y0_2/s0
*Get the estimated ATE
gen norm=y1-y0
gen norm2=y1_3-y0_3
*Same Results for ht:ht2 and norm:norm2
sum ht ht2 norm norm2
```

(140 real changes made)

(13,820 real changes made)

(16,177 real changes made)

(16,177 real changes made)

Variable	Obs	Mean	Std. Dev.	Min	Max
ht	16,177	-11876.79	127578.8	-99232.64	1.19e+07
ht2	16,177	-11876.79	0	-11876.79	-11876.79
norm	16,177	-7238.14	332633.6	-99127.2	3.11e+07
norm2	16,177	-7238.14	0	-7238.14	-7238.14

ATE under non-normalized weights is -\$11,876. ATE under normalized weights is -\$7,238.

Why are they so much different than the experimental research design estimates? Given that the mean $1 - p(x)$ is close to 1, a lot of weight is given to Y^0 . What we need to do is trim and recalculate.

2.1.2 Trim the data

We will drop our calculated variables and then trim our p -scores.

```
drop d1 d0 s1 s0 y1 y0 ht norm
drop if pscore <= 0.1
drop if pscore >= 0.9
```

We will recalculate the non-normalized weights using trimmed data.

```

gen d1=treat/pscore
gen d0=(1-treat)/(1-pscore)
egen s1=sum(d1)
egen s0=sum(d0)

gen y1=treat*re78/pscore
gen y0=(1-treat)*re78/(1-pscore)
gen ht=y1-y0

```

We will recalculate the normalized weights using trimmed data.

```

replace y1=(treat*re78/pscore)/(s1/_N)
replace y0=((1-treat)*re78/(1-pscore))/(s0/_N)
gen norm=y1-y0
sum ht norm

```

(0 real changes made)

(0 real changes made)

(0 real changes made)

Variable	Obs	Mean	Std. Dev.	Min	Max
ht	361	2006.365	19656.33	-73545.45	131755.3
norm	361	1806.73	18967.72	-72480.23	126251.9

ATE under non-normalized weights is \$2,006, while the estimated ATE under normalized weights is \$1,807.

These are much closer to our experimental research design estimates.

2.2 Calculate *ATE* and *ATT* using *teffects*

Luckily, we have a built-in Stata command with *teffects ipw*, so we don't need to manually calculate the *ATE* everytime. (However, it is a good exercise to see how Stata calculate the estimate). The *teffects* command will also calculate our standard errors, as well.

We will reload our data, trim the data, and use the *teffects ipw* command. We will also trim our data, since we know that there is a lack of overlap at the tail ends of the estimated *p*-scores.

```

* Reload Data
use https://github.com/scunning1975/mixtape/raw/master/nsw_mixtape.dta, clear
drop if treat==0

*Now merge in the CPS controls from footnote 2 of Table 2 (Dehejia and Wahba 2002)
append using https://github.com/scunning1975/mixtape/raw/master/cps_mixtape.dta
gen agesq=age*age
gen agecube=age*age*age
gen edusq=educ*educ
gen u74 = 0 if re74!=.
replace u74 = 1 if re74==0
gen u75 = 0 if re75!=.
replace u75 = 1 if re75==0
gen interaction1 = educ*re74
gen re74sq=re74^2
gen re75sq=re75^2
gen interaction2 = u74*hisp

* Now estimate the propensity score
logit treat age agesq agecube educ edusq marr nodegree black hisp re74 re75 u74 u75 in
predict pscore

* Checking mean propensity scores for treatment and control groups
sum pscore if treat==1, detail
sum pscore if treat==0, detail

* Trimming the propensity score
drop if pscore <= 0.1
drop if pscore >= 0.9

```

With the *teffects* command, we need to specify *ipw* for inverse probability weights and we need to use the option *logit* to tell Stata to calculate the *p*-scores using a logistic regression. In addition, we will use the option *osample* to look for any overlap between treatment and comparison. Finally, we will use the option *atet* to specify that we want to estimate the *ATT*.

```

teffects ipw (re78) (treat age agesq agecube educ edusq marr nodegree ///
black hisp re74 re75 u74 u75 interaction1, logit), osample(overlap) atet
capture drop overlap

```

```

Iteration 0:   EE criterion = 1.392e-21
Iteration 1:   EE criterion = 2.898e-23

```

```

Treatment-effects estimation              Number of obs      =      361

```



```

Estimator      : inverse-probability weights
Outcome model  : weighted mean
Treatment model: logit

```

	re78	Coef.	Robust Std. Err.	z	P> z	[95% Conf. Interval]	
ATET	treat (1 vs 0)	2646.866	892.2393	2.97	0.003	898.1089	4395.623
POmean	treat 0	3791.931	514.4727	7.37	0.000	2783.582	4800.279

The ATT is estimated at \$2,647

Before we calculate the *ATE*, we need to rescale the dependent variable for concavity of logit. We will do so by dividing re78 by 1000 dollars

The ATE is estimated at \$1,611

```

gen re78_scaled = re78/1000
teffects ipw (re78_scaled) (treat age agesq agecube educ edusq marr nodegree ///
black hisp re74 re75 u74 u75 interaction1, logit), osample(overlap) ate

```

```

command quietly is unrecognized
r(199);

```

```

r(199);

```


Chapter 3

Doubly Robust Estimator

Next we will implement our doubly robust estimator. We are combining our regression adjusted outcome model and our IPW using p -scores. We have two chances to get the correct specification with the doubly robust estimator.

$$ATE_{DR} = [E[Y^1] - E[Y^0]] + E \left[\frac{(Y - E[Y^1]) * D}{p(x)} - \frac{(Y - E[Y^0])(1 - D)}{(1 - p(x))} \right]$$

The left-hand side is our OLS-based model and the right-hand side is our Inverse Probability Weighted Model. If we specified the OLS model correctly, our second term equals zero.

First, we will reload our data and do the trimming.

```
* Reload Data
use https://github.com/scunning1975/mixtape/raw/master/nsw_mixtape.dta, clear
drop if treat==0

*Now merge in the CPS controls from footnote 2 of Table 2 (Dehejia and Wahba 2002)
append using https://github.com/scunning1975/mixtape/raw/master/cps_mixtape.dta
gen agesq=age*age
gen agecube=age*age*age
gen edusq=educ*educ
gen u74 = 0 if re74!=.
replace u74 = 1 if re74==0
gen u75 = 0 if re75!=.
replace u75 = 1 if re75==0
gen interaction1 = educ*re74
gen re74sq=re74^2
gen re75sq=re75^2
```

```

gen interaction2 = u74*hisp

* Now estimate the propensity score
logit treat age agesq agecube educ edusq marr nodegree black hisp re74 re75 u74 u75 in
predict pscore

* Checking mean propensity scores for treatment and control groups
sum pscore if treat==1, detail
sum pscore if treat==0, detail

* Trimming the propensity score
drop if pscore <= 0.1
drop if pscore >= 0.9

```

3.1 Calculate the *ATE* and *ATT* using Doubly Robust Estimator

Our Doubly Robust Estimator using *teffects* includes *teffects ipwra* (*OUTCOME MODEL*) (*TREATMENT MODEL*) In the first set of paratheses, we will use the variables to estimate the regression adjustment model or outcome model. In the second set of paratheses we will estimate the treatment model or the model for to estimate our *p*-scores.

Next, we will calculate the *ATT*. Note: We have to rescale for the concavity of the logit.

```

gen re78_scaled = re78/1000

teffects ipwra (re78_scaled age agesq agecube educ edusq ///
marr nodegree black hisp re74 re75 u74 u75 interaction1) (treat age agesq agecube educ
black hisp u74 u75 interaction1, logit), atet

```

```

Iteration 0:   EE criterion = 2.996e-21
Iteration 1:   EE criterion = 6.603e-29

```

```

Treatment-effects estimation              Number of obs      =          361
Estimator      : IPW regression adjustment
Outcome model   : linear
Treatment model: logit

```

		Robust			
	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]
re78_scaled					

3.1. CALCULATE THE ATE AND ATT USING DOUBLY ROBUST ESTIMATOR29

ATET							
treat							
(1 vs 0)		2.565509	.8702429	2.95	0.003	.8598642	4.271154

P0mean							
treat							
0		3.873287	.4696332	8.25	0.000	2.952823	4.793752

We find that our estimate *ATT* is \$2,566.

Next we will estimate the *ATE*, and we use the option *ate* to specify that we want to estimate the *ATE*.

```
teffects ipwra (re78_scaled age agesq agecube educ edusq ///
marr nodegree black hisp re74 re75 u74 u75 interaction1) (treat age agesq agecube educ edusq marr
black hisp u74 u75 interaction1, logit), ate
```

```
Iteration 0: EE criterion = 2.996e-21
Iteration 1: EE criterion = 1.056e-28
```

```
Treatment-effects estimation          Number of obs    =          361
Estimator       : IPW regression adjustment
Outcome model   : linear
Treatment model : logit
```

			Robust				
re78_scaled		Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	

ATE							
treat							
(1 vs 0)		1.608922	.6988993	2.30	0.021	.2391043	2.978739

P0mean							
treat							
0		4.297229	.3749753	11.46	0.000	3.562291	5.032167

We find that our estimate *ATE* is \$1,609.

Cunningham, S. (2021). Causal Inference: The Mixtape. Yale University Press.

Huber, M. (2023). Causal Analysis: Impact Evaluation and Causal Machine Learning with Applications in R. The MIT Press.

Huntington-Klein, N. (2022). The Effect: An Introduction to Research Design and Causality. CRC Press.